
Antes de empezar
Este cuadernillo es el paso siguiente a Excel básico: está pensado para quien ya carga datos y hace sumas, y necesita

cruzar información, contar o sumar con condiciones, limpiar textos, calcular fechas y resumir una base grande en

pocos clics. Los ejemplos usan Excel en español con la configuración de Argentina: la coma es el separador decimal y

los argumentos se separan con punto y coma (;). Algunas funciones (BUSCARX, FILTRAR, UNICOS) necesitan Excel

2021 o posterior, Microsoft 365 o Excel en la web; otras (SI.CONJUNTO, CONCAT, UNIRCADENAS) están desde Excel

2019.

CÓMO  APROVECHARLO

Armá cada ejemplo en una hoja nueva y cambiá los datos para ver cómo responde la fórmula.

Hacé los ejercicios del final de cada tema antes de mirar las respuestas.

Cuando termines, rendí la evaluación Excel: fórmulas y tablas dinámicas en testing-us.com.

TIP Si recién empezás con Excel, primero repasá el cuadernillo de Excel básico de Testing-US: este parte de lo que

se ve ahí.

Contenido

1 Referencias, nombres y errores

Relativas, absolutas y mixtas, otras hojas, nombres, tablas y errores.

2 Funciones de búsqueda

BUSCARV exacta y por tramos, BUSCARX, INDICE y COINCIDIR.

3 Funciones lógicas y condicionales

SI, Y, O, SI.CONJUNTO, CONTAR.SI.CONJUNTO, SUMAR.SI.CONJUNTO y SI.ERROR.

4 Texto, fechas y matrices dinámicas

EXTRAE, ESPACIOS, TEXTO, DIAS.LAB, FIN.MES, UNICOS y FILTRAR.

5 Tablas dinámicas

Origen de datos, áreas, resúmenes, porcentajes, agrupar, filtrar y actualizar.

Testing-US · Excel: fórmulas y tablas dinámicas · Cuadernillo de estudio 1



1
TEMA  1

Referencias, nombres y errores

LO  QUE  VAS  A  APRENDER

Elegir entre referencias relativas, absolutas y mixtas.

Usar celdas de otras hojas y nombres definidos.

Trabajar con tablas de Excel y referencias

estructuradas.

Reconocer cada error y su causa.

Referencias relativas, absolutas y mixtas

Una referencia relativa ( A1 ) cambia al copiar la fórmula: si en C2 dice =A2*B2  y la copiás a C3, pasa a ser

=A3*B3 .

Una referencia absoluta ( $A$1 ) no cambia al copiar. Se usa para un valor fijo, por ejemplo una tasa o un tipo de

cambio en una sola celda: =B2*$F$1 .

Las referencias mixtas fijan solo una parte: $A1  fija la columna (la fila cambia) y A$1  fija la fila (la columna

cambia). Sirven para tablas de doble entrada, como una tabla de multiplicar: =$A2*B$1 .

Con el cursor sobre la referencia, la tecla F4 recorre los cuatro tipos: A1  → $A$1  → A$1  → $A1  → A1 .

A B C D E F

1

2

3

4

5

6

× 1

1 =$A2*B$1 2 3 4 5

2

2 2 4 6 8 10

3

3 3 6 9 12 15

4

4 4 8 12 16 20

5

5 5 10 15 20 25

En E5 la fórmula copiada es

=$A5*E$1

$A5: la columna A queda fija,

la fila cambia (5)

E$1: la fila 1 queda fija,

la columna cambia (E)

Figura 1.1  Una sola fórmula con referencias mixtas llena toda la tabla.

Otras hojas, otros libros y nombres

Para usar una celda de otra hoja: =Ventas!B2 . Si el nombre de la hoja tiene espacios, va entre comillas simples:

='Ventas 2026'!B2 .

Se le puede poner un nombre a una celda o a un rango (cuadro de nombres, a la izquierda de la barra de fórmulas, o

Fórmulas > Administrador de nombres). Así, =B2*IVA  se entiende mejor que =B2*$F$1 . Los nombres funcionan

como referencias absolutas.

Tablas de Excel

Con Insertar > Tabla un rango se convierte en una tabla: tiene encabezados con filtros, filas con colores alternados

y se amplía sola cuando se agregan filas debajo.

Las fórmulas de una tabla usan referencias estructuradas: =[@Cantidad]*[@Precio]  (la fila actual) o

=SUMA(Ventas[Importe])  (toda la columna). Una fórmula escrita en una columna se copia sola a toda la columna.

La fila de totales agrega al final un total por columna (suma, promedio, recuento…).

TIP Convertí en tabla toda base que vaya a crecer: las fórmulas, los formatos y las tablas dinámicas que dependen

de ella se actualizan con las filas nuevas.

Errores frecuentes

#N/D : una búsqueda no encontró el valor.

#¡DIV/0! : se dividió por cero o por una celda vacía.

Testing-US · Excel: fórmulas y tablas dinámicas · Cuadernillo de estudio 2



#¿NOMBRE? : Excel no reconoce un nombre, casi siempre una función mal escrita ( =SUMAA(A1:A5) ) o un texto sin

comillas.

#¡VALOR! : se usó un tipo de dato equivocado, como sumar con +  un texto que no es un número (por ejemplo,

"N/A"). Una celda vacía, en cambio, cuenta como 0 y no da error.

########  no es un error: la columna es angosta para mostrar el número o la fecha (se arregla ensanchándola).

#¡REF! : la fórmula apunta a una celda que ya no existe (por ejemplo, porque se eliminó la columna).

#¡DESBORDAMIENTO! : una fórmula que devuelve varios resultados no tiene lugar libre para mostrarlos.

Fórmulas > Evaluar fórmula muestra el cálculo paso a paso, y Rastrear precedentes dibuja flechas desde las

celdas que usa una fórmula hasta la celda de la fórmula.

#N/D

La búsqueda no encontró el valor

#¡DIV/0!

Se dividió por cero o por una celda vacía

#¿NOMBRE?

Función o nombre mal escrito

#¡VALOR!

Tipo de dato equivocado (un texto en una cuenta)

#¡REF!

La celda a la que apuntaba ya no existe

#¡DESBORDAMIENTO!

No hay lugar para mostrar todos los resultados

Figura 1.2  Los errores más comunes y qué los causa.

Para practicar

1. En C2 está =A2*$E$1. ¿Qué fórmula queda al copiarla a

C5?

2. ¿Qué referencia usarías para fijar solo la fila 3?

3. ¿Cómo se escribe una referencia a A1 de la hoja Datos

2026?

4. Eliminaste una columna y aparece #¡REF!. ¿Qué pasó?

LAS  RESPUESTAS  ESTÁN  AL  FINAL  DEL  CUADERNILLO.

2
TEMA  2

Funciones de búsqueda

LO  QUE  VAS  A  APRENDER

Usar BUSCARV con coincidencia exacta y por tramos.

Buscar en cualquier dirección con BUSCARX.

Combinar INDICE y COINCIDIR.

Encontrar la causa de un #N/D.

BUSCARV

=BUSCARV(valor_buscado; matriz; columna; [rango])  busca un valor en la primera columna de la matriz y devuelve el

dato de la columna indicada, en la misma fila.

Ejemplo: =BUSCARV(A2; Productos!$A$2:$C$500; 3; FALSO)  busca el código de A2 en la hoja Productos y devuelve la

3.ª columna (por ejemplo, el precio).

El último argumento en FALSO (o 0) pide coincidencia exacta: es lo que se usa casi siempre con códigos, DNI o

nombres. Si no lo encuentra, da #N/D .

Con VERDADERO u omitido hace una coincidencia aproximada: devuelve el valor más grande que sea menor o

igual al buscado. Sirve para tramos (comisiones, escalas de notas, categorías de impuestos) y exige que la primera

columna esté ordenada de menor a mayor.

Limitaciones: solo busca en la primera columna y solo devuelve datos a la derecha; el número de columna es fijo, así

que si se inserta una columna en el medio, la fórmula puede devolver otro dato.

Conviene fijar la matriz con $  para poder copiar la fórmula hacia abajo.

BUSCARH hace lo mismo pero busca en la primera fila y devuelve una fila de abajo.

A B C

1

2

3

4

5

Código Producto Precio

A-101 Yerba 1 kg 4.800

A-205 Café 500 g 7.950

B-310 Azúcar 1 kg 1.350

B-402 Aceite 900 ml 3.200

=BUSCARV("A-205"; A2:C5; 3; FALSO)

1 2

1. Busca A-205 en la primera columna

(con FALSO: coincidencia exacta)

2. Devuelve la columna 3 de esa fila

Resultado: 7.950

No puede devolver columnas a la izquierda.

Figura 2.1  BUSCARV busca en la primera columna y devuelve otra columna de la misma fila.

BUSCARX

=BUSCARX(valor_buscado; matriz_buscada; matriz_devuelta; [si_no_se_encuentra]; [modo_coincidencia]; [modo_búsqueda])

Ejemplo: =BUSCARX(A2; Productos!A:A; Productos!C:C; "Sin precio") .

Por defecto busca la coincidencia exacta; no hace falta ordenar.

Puede devolver una columna que esté a la izquierda de la buscada, y no se rompe si se insertan columnas.

El cuarto argumento define qué mostrar si no encuentra el valor (en vez de #N/D ).

Con el modo de búsqueda -1  busca desde el final hacia arriba: devuelve la última aparición.

INDICE y COINCIDIR

=COINCIDIR(valor; rango; 0)  devuelve la posición del valor dentro del rango (con 0, coincidencia exacta).

=COINCIDIR("Pera"; {"Manzana";"Pera";"Uva"}; 0)  da 2.

=INDICE(rango; fila; [columna])  devuelve el valor que está en esa posición. =INDICE(C2:C10; 4)  devuelve el 4.º valor

de C2:C10.

Juntas reemplazan a BUSCARV en cualquier versión de Excel y también buscan hacia la izquierda:

=INDICE(C2:C100; COINCIDIR(F2; A2:A100; 0)) .

Con dos COINCIDIR se busca en una tabla de doble entrada:

=INDICE(B2:M20; COINCIDIR(producto; A2:A20; 0); COINCIDIR(mes; B1:M1; 0)) .

=INDICE(C2:C100; COINCIDIR(F2; A2:A100; 0))

Paso 1 · COINCIDIR

Busca F2 en A2:A100

y devuelve su posición:

por ejemplo, 37

Paso 2 · INDICE

Devuelve el valor 37.º

del rango C2:C100

(puede estar a la izquierda)

Figura 2.2  COINCIDIR da la posición; INDICE devuelve el valor de esa posición.

Buenas prácticas

Si una búsqueda da #N/D  aunque el dato "está", revisá espacios de más (se limpian con ESPACIOS) o números

guardados como texto.

Para mostrar un mensaje en lugar del error: =SI.ERROR(BUSCARV(...); "No encontrado") .

OJO Con códigos, DNI o nombres usá siempre coincidencia exacta (FALSO en BUSCARV). La coincidencia

aproximada con una lista desordenada devuelve datos equivocados sin mostrar ningún error.

Para practicar

1. ¿Qué devuelve BUSCARV con FALSO si el valor no está

en la lista?

2. Tramos 0 → 5 %, 100.000 → 8 %, 500.000 → 10 %. ¿Qué

tramo corresponde a 499.999?

3. Escribí una fórmula que busque el DNI de F2 en C:C y

devuelva el nombre de A:A.

4. ¿Qué dos causas comunes hacen que un BUSCARV dé

#N/D aunque el dato parezca estar?

LAS  RESPUESTAS  ESTÁN  AL  FINAL  DEL  CUADERNILLO.

3
TEMA  3

Funciones lógicas y condicionales

LO  QUE  VAS  A  APRENDER

Escribir condiciones con SI, Y y O.

Ordenar bien varias condiciones (SI anidados y

SI.CONJUNTO).

Contar y sumar con una o varias condiciones.

Escribir criterios con comparaciones, celdas y

comodines.

SI, Y, O

=SI(prueba; valor_si_verdadero; valor_si_falso) . Ejemplo: =SI(B2>=6; "Aprobado"; "Desaprobado") .

Los textos van entre comillas; los números y las referencias, sin comillas.

=Y(condición1; condición2; …)  es VERDADERO solo si todas se cumplen; =O(…)  es VERDADERO si se cumple al

menos una. NO(…)  invierte el resultado.

Se combinan: =SI(Y(B2>=6; C2>=75%); "Regular"; "Libre") .

Para comparar: = , <>  (distinto), > , < , >= , <= .

Para saber si una celda está vacía se compara con dos comillas: =SI(A2=""; "Falta"; "OK") .

Varias condiciones: SI anidados y SI.CONJUNTO

Un SI dentro de otro evalúa en orden: =SI(B2>=9; "Excelente"; SI(B2>=6; "Aprobado"; "Desaprobado")) . Hay que poner

primero la condición más exigente.

=SI.CONJUNTO(B2>=9; "Excelente"; B2>=6; "Aprobado"; VERDADERO; "Desaprobado")  hace lo mismo sin anidar:

devuelve el valor de la primera condición verdadera. El último par con VERDADERO funciona como "en cualquier

otro caso"; si ninguna condición se cumple y no hay ese par, da #N/D .

Bien: la más exigente primero

¿nota >= 9? → Excelente

¿nota >= 6? → Aprobado

si no → Desaprobado

9,5 da Excelente ✓

Mal: la más amplia primero

¿nota >= 6? → Aprobado

¿nota >= 9? → Excelente

si no → Desaprobado

9,5 da Aprobado ✗ (nunca llega)

Figura 3.1  Con condiciones que se superponen, la más exigente va primero.

Contar y sumar con condiciones

=CONTAR.SI(rango; criterio) : cuenta las celdas que cumplen un criterio. =CONTAR.SI(C2:C200; "Neuquén") .

=SUMAR.SI(rango; criterio; [rango_suma]) : suma los valores de rango_suma cuyas filas cumplen el criterio.

=SUMAR.SI(C2:C200; "Neuquén"; E2:E200) .

Con varias condiciones: =CONTAR.SI.CONJUNTO(rango1; criterio1; rango2; criterio2; …)  y

=SUMAR.SI.CONJUNTO(rango_suma; rango1; criterio1; rango2; criterio2; …) . Ojo: en SUMAR.SI.CONJUNTO el rango a

sumar va primero.

PROMEDIO.SI , PROMEDIO.SI.CONJUNTO , MAX.SI.CONJUNTO  y MIN.SI.CONJUNTO  funcionan igual.

Todas las condiciones de las funciones .CONJUNTO se tienen que cumplir a la vez (como un Y).

A B C D

1

2

3

4

5

6

Fecha Provincia Rubro Importe

03/03 Neuquén Bebidas 120.000

05/03 Río Negro Bebidas 85.000

08/03 Neuquén Almacén 64.000

12/03 Neuquén Bebidas 47.500

15/03 Mendoza Bebidas 99.000

=SUMAR.SI.CONJUNTO(D2:D6; B2:B6; "Neuquén"; C2:C6; "Bebidas")

Suma solo las filas que

cumplen las DOS condiciones:

120.000 + 47.500

= 167.500

Figura 3.2  SUMAR.SI.CONJUNTO suma las filas que cumplen todas las condiciones.

Cómo escribir los criterios

Un texto o número exacto: "Neuquén" , 100 .

Con comparación, entre comillas: ">100" , "<>Anulada" .

Con una celda, se une con & : ">"&H1  (mayor que el valor de H1).

Comodines: *  reemplaza cualquier cantidad de caracteres y ?  uno solo. "Ana*"  incluye "Ana", "Ana López" y

"Anabel".

Los criterios no distinguen mayúsculas de minúsculas.

SI.ERROR

=SI.ERROR(valor; valor_si_error)  devuelve el resultado de la fórmula o, si da cualquier error, el valor alternativo:

=SI.ERROR(A2/B2; 0) . Conviene usarlo solo cuando el error es esperable, para no esconder errores de verdad.

Para practicar

1. ¿Qué devuelve =Y(5>3; 2>4)?

2. Escribí una fórmula que cuente las facturas de B2:B500

que dicen "Pendiente".

3. Escribí el criterio "menor que el valor de K1".

4. ¿Qué cuenta =CONTAR.SI(A:A; "*SRL")?

LAS  RESPUESTAS  ESTÁN  AL  FINAL  DEL  CUADERNILLO.

4
TEMA  4

Texto, fechas y matrices dinámicas

LO  QUE  VAS  A  APRENDER

Cortar, limpiar y unir textos.

Calcular días, días hábiles y fines de mes.

Mostrar fechas dentro de un texto.

Usar UNICOS, ORDENAR y FILTRAR.

Funciones de texto

IZQUIERDA(texto; n)  y DERECHA(texto; n)  devuelven los primeros o los últimos n caracteres. EXTRAE(texto; inicio; n)
devuelve n caracteres desde una posición: =EXTRAE("AR-2026-015"; 4; 4)  da "2026".

LARGO(texto)  cuenta los caracteres, incluidos los espacios.

ESPACIOS(texto)  quita los espacios del principio y del final, y deja uno solo entre palabras.

MAYUSC , MINUSC  y NOMPROPIO  cambian mayúsculas: =NOMPROPIO("ana pérez")  da "Ana Pérez".

HALLAR(buscado; texto)  y ENCONTRAR(buscado; texto)  devuelven la posición de un texto dentro de otro; HALLAR no

distingue mayúsculas y ENCONTRAR sí.

SUSTITUIR(texto; viejo; nuevo)  reemplaza un texto por otro.

Para unir: el operador &  ( =A2&" "&B2 ), CONCAT(A2; " "; B2)  o UNIRCADENAS(", "; VERDADERO; A2:A10) , que pone

un separador entre cada valor y puede ignorar las celdas vacías.

TEXTO(valor; formato)  convierte un número o una fecha en texto con formato:

="Vence el "&TEXTO(B2; "dd/mm/aaaa") .

Datos > Texto en columnas separa un texto en varias columnas (por ejemplo, por comas); Relleno rápido completa

una columna imitando un ejemplo.

A

1

R

2

-

3

2

4

0

5

2

6

6

7

-

8

0

9

1

10

5

11

=EXTRAE("AR-2026-015"; 4; 4) → "2026"

Desde la posición 4, toma 4 caracteres.

Figura 4.1  EXTRAE cuenta las posiciones desde 1.

Fechas

Excel guarda las fechas como números de serie (cada día suma 1). Por eso, restar dos fechas da la cantidad de días:

=B2-A2 .

HOY()  devuelve la fecha actual y AHORA()  la fecha y la hora; se actualizan solas cada vez que se recalcula la hoja.

FECHA(año; mes; día)  arma una fecha; AÑO , MES  y DIA  extraen cada parte.

DIAS(fecha_final; fecha_inicial)  da los días entre dos fechas (ojo con el orden: primero la final).

DIAS.LAB(inicio; fin; [feriados])  cuenta los días hábiles (lunes a viernes) entre dos fechas, incluidas ambas, y

descuenta los feriados que se le pasen en un rango.

FIN.MES(fecha; meses)  da el último día del mes: =FIN.MES(A2; 0)  es el fin del mismo mes. FECHA.MES(fecha; meses)
suma meses.

01/03/2026

número de serie 46082

02/03/2026

número de serie 46083

31/03/2026

número de serie 46112

=B2-A2 → 31/03/2026 − 01/03/2026 = 30 días

Por eso se pueden restar fechas y sumarles días.

Figura 4.2  Detrás de cada fecha hay un número de serie.

Matrices dinámicas

En Microsoft 365 y Excel 2021, algunas fórmulas devuelven varios resultados que se desbordan en las celdas de al

lado.

=UNICOS(A2:A500)  devuelve la lista sin repetidos; =ORDENAR(A2:C500; 3; -1)  ordena por la 3.ª columna de mayor a

menor; =FILTRAR(A2:C500; C2:C500>100000; "Sin datos")  devuelve solo las filas que cumplen la condición.

Si alguna celda donde se tiene que desbordar el resultado está ocupada, aparece #¡DESBORDAMIENTO! . Para

referirse a todo el resultado se usa el símbolo # : =CONTARA(E2#) .

TIP Las matrices dinámicas no se escriben celda por celda: escribí la fórmula una sola vez y dejá libres las celdas

de abajo y de la derecha.

Para practicar

1. ¿Qué devuelve =DERECHA("AR-2026-015"; 3)?

2. ¿Qué fórmula une A2 y B2 con una coma y un espacio en

el medio?

3. ¿Cuántos días hábiles hay del lunes 2 al viernes 6 de

marzo de 2026, sin feriados?

4. ¿Qué pasa si una celda debajo de una fórmula UNICOS

está ocupada?

LAS  RESPUESTAS  ESTÁN  AL  FINAL  DEL  CUADERNILLO.

5
TEMA  5

Tablas dinámicas

LO  QUE  VAS  A  APRENDER

Preparar los datos de origen.

Armar la tabla con Filas, Columnas, Valores y Filtros.

Cambiar el resumen, mostrar porcentajes y agrupar

fechas.

Filtrar con segmentaciones y actualizar.

Para qué sirven

Una tabla dinámica resume miles de filas en pocos segundos: total de ventas por vendedor y mes, cantidad de

reclamos por sucursal, promedio por categoría. No modifica los datos originales.

Antes de crearla

Los datos tienen que estar en forma de lista: una fila de encabezados únicos, un registro por fila, sin filas ni

columnas en blanco y sin celdas combinadas.

Conviene que el origen sea una tabla de Excel: así, las filas nuevas se incluyen solas al actualizar.

Crearla y armarla

Insertar > Tabla dinámica, elegir el origen y dónde ubicarla (conviene una hoja nueva).

En el panel Campos de tabla dinámica se arrastran los campos a cuatro áreas: Filas, Columnas, Valores y Filtros.

En Valores, un campo numérico se resume por defecto con Suma; un campo de texto, con Recuento (cuenta los

registros). Con Configuración de campo de valor se cambia a Promedio, Máx, Mín, etc.

Mostrar valores como permite ver, por ejemplo, el % del total general o la diferencia con el mes anterior, sin

fórmulas.

Un campo de fechas se puede agrupar por meses, trimestres o años (clic derecho > Agrupar).

Un campo calculado crea una columna nueva a partir de otras (por ejemplo, Importe × 21 %).

1. Datos (una fila por venta)

A B C

1

2

3

4

5

6

Vendedor Mes Importe

Paz Ene 150

Ruiz Ene 90

Paz Feb 120

Ruiz Feb 200

Paz Feb 60

2. Campos

Filas
Vendedor

Columnas
Mes

Valores
Suma de Importe

Filtros
(vacío)

3. Resumen

A B C D

1

2

3

4

Ene Feb Total

Paz 150 180 330

Ruiz 90 200 290

Total 240 380 620

Figura 5.1  De una lista de ventas a un resumen por vendedor y mes.

Filtrar y explorar

El área Filtros filtra todo el informe; también se puede filtrar desde los encabezados de filas y columnas.

Las segmentaciones de datos son botones para filtrar a la vista (Analizar tabla dinámica > Insertar segmentación de

datos); la escala de tiempo hace lo mismo con fechas.

Hacer doble clic en un número de la tabla dinámica abre una hoja nueva con las filas de origen que lo forman.

Un gráfico dinámico se actualiza junto con la tabla dinámica.

Actualizar

La tabla dinámica no se actualiza sola cuando cambian los datos: hay que usar Actualizar (clic derecho >

Actualizar, o Datos > Actualizar todo).

Si se agregaron filas fuera del rango de origen y el origen no es una tabla, hay que cambiar el origen de datos.

OJO Si presentás un informe con una tabla dinámica, actualizala antes: no cambia sola cuando cambian los datos.

Para practicar

1. ¿Dónde ubicás los campos para ver el total de reclamos

por sucursal y por mes?

2. Arrastraste el campo DNI a Valores. ¿Qué vas a ver?

3. ¿Cómo ves el porcentaje que representa cada rubro

sobre el total?

4. Agregaste ventas nuevas a la base y no aparecen. ¿Qué

dos cosas revisás?

LAS  RESPUESTAS  ESTÁN  AL  FINAL  DEL  CUADERNILLO.

Testing-US · Excel: fórmulas y tablas dinámicas · Cuadernillo de estudio 11



✓
PARA  CONTROLAR

Respuestas de los ejercicios

Tema 1

1. =A5*$E$1: la referencia relativa cambia y la absoluta queda fija.

2. Una mixta con $ delante del número: A$3.

3. ='Datos 2026'!A1 (con comillas simples por el espacio).

4. La fórmula apuntaba a una celda de esa columna, que ya no existe.

Tema 2

1. El error #N/D.

2. 8 %: es el tramo más alto que no supera 499.999.

3. =BUSCARX(F2; C:C; A:A) o =INDICE(A:A; COINCIDIR(F2; C:C; 0)).

4. Espacios de más y números guardados como texto (o al revés).

Tema 3

1. FALSO: la segunda condición no se cumple.

2. =CONTAR.SI(B2:B500; "Pendiente").

3. "<"&K1

4. Las celdas que terminan en SRL.

Tema 4

1. 015.

2. =A2&", "&B2 (o =CONCAT(A2; ", "; B2)).

3. 5: DIAS.LAB cuenta el día de inicio y el final.

4. Aparece #¡DESBORDAMIENTO! hasta que se libere el lugar.

Tema 5

1. Sucursal en Filas, Mes (agrupado) en Columnas y Reclamo en Valores.

2. Un recuento (o una suma de números de DNI, que no sirve): cambiá el resumen a Recuento.

3. Configuración de campo de valor > Mostrar valores como > % del total general.

4. Actualizar la tabla dinámica y revisar que el origen incluya las filas nuevas (mejor si es una tabla).

Validá lo que aprendiste
Rendí la evaluación Excel: fórmulas y tablas dinámicas en testing-us.com. Es gratis: en

cada intento te tocan 10 preguntas al azar, dos de cada tema de este cuadernillo. Tenés

15 minutos y aprobás con 70%.

Si aprobás, recibís un certificado con código de verificación que podés sumar a tu

currículum y compartir con quien te contrate.

MIS  NOTAS

© 2026 Testing-US. Material de estudio propio. Se permite imprimirlo y compartirlo sin modificaciones y sin fines comerciales.

Referencia: ayuda oficial de Excel (Microsoft, support.microsoft.com). Los textos y dibujos de este cuadernillo son propios y esquemáticos; no

son capturas del programa. Microsoft y Excel son marcas registradas de Microsoft Corporation; se mencionan solo para identificar el

programa. Este material no está respaldado por Microsoft.

El certificado de Testing-US acredita la aprobación de una evaluación de conocimientos. No es un título oficial ni habilita para el ejercicio

profesional.

